CHAPTER 15

Discounted Cash Flow Valuation

There are many different ways to value a company, but one that makes use of a projection financial model is the discounted cash flow (DCF) valuation method. This is based on the concept that the value of an entity is based on the future cash flows it can generate. The value is arrived at when those cash flows are discounted (or present-valued) at a discount rate that represents the risk of those cash flows.

DCF is not limited to just valuing a company. It can be used to value a division of a company, a project, an asset—basically any enterprise that generates a cash flow. For this reason, we will use the term *entity* instead of *company* in describing the elements of a DCF valuation.

The elements of a DCF analysis are:

- The weighted average cost of capital
- The free cash flows
- The terminal value
- The enterprise value
- The equity value

15.1 THE FILE TO USE

Use *Chapter 15 DCF.xlsm* from the buildingfinancialmodels.com website. This file has the three completed worksheets from Chapter 12. Rather than giving you a blow-by-blow description of how to build the DCF worksheet, I have included a completed DCF sheet in the workbook to show the topics covered in this chapter. A good exercise is to create for yourself a duplicate DCF worksheet to familiarize yourself with the steps in building this.

15.2 WEIGHTED AVERAGE COST OF CAPITAL

Weighted average cost of capital (or WACC, pronounced "wack") is the blended cost of the equity and debt capital of the entity.

The cost of equity is the return required by equity holders for the risk of investing their capital into the company. The cost of debt is the return required by debt holders. However, because debt interest is tax deductible, the cost of debt required is really the after-tax cost of debt [interest rate % * (1 − tax rate %)]. The blended value is based on the portion of market value of the equity and market value of the debt in the entity's capitalization.

Market value of the equity is the number of shares outstanding multiplied by the current share price.

Market value of the debt is the sum of the present value of each year's interest discounted at the cost of debt, plus the present value of the repayment discounted at the cost of debt. If the market cost of debt is the same as the coupon yield, then the market value is the same as the book value. If the market cost is higher, then the market value is lower than the book value, and vice versa. In the absence of any information that would allow you to calculate the market value, use the book value of debt.

In an acquisition scenario where there is an acquirer and a target, and the target is being valued by DCF, the WACC to use in valuing the target is the target's WACC, not the acquirer's. The reason is that discount rate should reflect the risk of the cash flows; the risk in this case is attached to the cash flows of the target.

15.3 FREE CASH FLOW

The cash flows in a DCF are called "free cash flow" (FCF). Free cash flow represents the cash flows available to the potential buyer or investor after all the cash sources and all the necessary cash uses to keep the business operating are taken into account. FCF consists of:

- The operating earnings of the entity. For a company, this is the EBIT line. The rationale here is that EBIT reflects the flows generated by the operations of the company. EBIT is also before interest (due to the debt holders) and dividends (due to the investors), so it represents the flows to the company itself.
- Reduced by: taxes on the EBIT. The tax is an artificial number in that it is not the tax seen in the forecast income statement, which is based on the marginal tax rate times earnings before taxes (EBT). For the DCF, we apply the tax rate on EBIT, not EBT. Because taxes are a real cost, this is computed. The taxes are on EBIT, not EBITDA, because depreciation and amortization are tax-deductible.

  In DCF terminology, the EBIT after taxes is called EBIAT, or earnings before interest after taxes. A parallel terminology is operating profit for EBIT, and NOPAT, or net operating profit after taxes, for EBIAT. We will use the terms EBIT and EBIAT.
- Increased by: depreciation and amortization. These are noncash expenses, so they are added back. The depreciation and amortization amounts should be the tax basis amounts, not the book amounts. Depreciation on a tax-basis can be different from the book basis. Using cash-basis numbers for these two items means we do not have to worry about deferred taxes.
- Reduced by: any increase in net working capital. Any increase in net working capital is a use of funds. This is

why in the illustrations below an increase is shown as a negative number, a decrease as a positive number.
- Reduced by: capital expenditures.

Increases in net working capital and capital expenditures reduce the free cash flow. Such increases need to be taken into account in the calculation of FCF because they represent the cash that must be invested in order to sustain the operations.

The free cash flow of each projected year is discounted back to the valuation date at the WACC. There are two ways of discounting the free cash flows.

15.4 DISCOUNTING

Once we have the FCF, we have to *discount* each year's FCF to the valuation date. Discounting takes into account the time value of money, which is the idea that $100 available today is worth more than $100 available a year hence (it's the financial equivalent of the saying "a bird in hand is worth two in the bush"). Put another way, the $100 that you have to wait a year for to get is worth less than the $100 you can get today. The question is: By how much less? If I have a savings account with a 3% interest rate, I can put in some money now that in a year's time will be $100. The amount now would be $97.09 (from $100/1.03), and that will grow in value to $100 in a year's time. To carry the example further, to get a value of $100 in two years' time, the amount required now would be $96.26, or $100/(1.03)^2. The discounting factor has an exponent to indicate the number of years from today's date, that is, the number of times the 1.03 is applied from today's date.

That 3% is the discount rate. For our valuation discounting, the discount rate is the WACC, and we apply the discounting factors to each of the forecast year's FCF.

15.4.1 Discounting by the year-end convention

In year-end discounting, the free cash flows are recognized at the end of each year. Let's say the valuation date is January 1, 2020, and the discount rate is 10%. We have an FCF of $100 per month for a total $1,200, all of which is recognized at December 31, 2020. The discounted value at the valuation date is $1,091, or $1,200/(1 + 10%).

15.4.2 Discounting by the midyear convention

In this approach, each year's free cash flow is recognized at the midpoint of the year. The thinking is that since the free cash flows are generated each month-end through the year, there is a good rationale to recognize them as an average at the midpoint of the year. In this method, if a valuation date is January 1, 2020, and the discount rate is 10%, then the free cash flow of $1,200 for the year 2020 is discounted by only half a year. The discounted value is $1,144 ($1,200/(1 + 10%)^0.5). The midyear approach always results in a higher valuation than the year-end convention (Figure 15-1).

FIGURE 15-1

Comparing period-end with midperiod discounting



15.4.3 Valuation at December 1, 2020

Figure 15-2 shows the discounting calculations when the valuation date is set to December 31, 2020, which is the same as January 1, 2021. This figure, and the one below it, are in the *Chapter 15 DCF.xlsm* workbook, in the "Discount years" tab.

FIGURE 15-2

Valuation at the beginning of 2021

	B	C	D	E	F
1	**Valuation date at December 31, 2020**				
2	WACC	10.0%			
3					
4	**Year-end discounting**	**2021**	**2022**	**2023**	**Formula in column C**
5	Free cash flow	1,200	1,260	1,323	
6	Discounting years	*1.000*	*2.000*	*3.000*	1 in 1st Year, then increase of 1 per year
7	Discounting factor	*0.909*	*0.826*	*0.751*	=1/(1+$C$2)^C6
8	**Present value of FCF**	**3,126**			=SUMPRODUCT(C5:E5,C7:E7)
9					
10	**Mid-year discounting**	**2021**	**2022**	**2023**	**Formula in column C**
11	Free cash flow	1,200	1,260	1,323	
12	Discounting years	*0.500*	*1.500*	*2.500*	0.5 in 1st Year, then increase of 1 per year
13	Discounting factor	*0.953*	*0.867*	*0.788*	=1/(1+$C$2)^C12
14	**Present value of FCF**	**3,279**			=SUMPRODUCT(C11:E11,C13:E13)
15					

15.4.3.1 Year-end discounting

The year-end discounting years row (row 6) is straightforward, counting the years for discounting. These numbers are used as exponents in the discounting factors (row 7). The present value of the free cash flow is in C8 by the SUMPRODUCT function, which multiplies the free cash flow range by the discounting factor range.

15.4.3.2 Midyear discounting

The midyear discounting years row (row 12) starts with 0.5, because in the first year, the FCF is recognized at the June 30, 2021, midpoint of the full year. The second year's discount years count is 1.5 because June 30, 2022, is 1.5 years away from December 31, 2020.

15.4.4 Valuation at March 31, 2021

With the valuation date set at March 31, 2021, or one-quarter into the year, the discounting calculations require a little more attention (Figure 15-3).

FIGURE 15-3

Valuation at the end of the first quarter of 2021

	A	B	C	D	E	F
21		**Valuation date at March 31, 2021**				
22		WACC	10.0%			
23						
24		**Year-end discounting**	**2021**	**2022**	**2023**	**Formula in column C**
25		Free cash flow	900	1,260	1,323	
26		Discounting years	0.750	1.750	2.750	0.75 in 1st Year, then increase of 1 per year
27		Discounting factor	0.931	0.846	0.769	=1/(1+$C$2)^C26
28		**Present value of FCF**	**2,922**			=SUMPRODUCT(C25:E25,C27:E27)
29						
30		**Mid-year discounting**	**2021**	**2022**	**2023**	**Formula in column C**
31		Free cash flow	900	1,260	1,323	
32		Discounting years	0.375	1.250	2.250	0.75/2 in 1st Year, 0.75+0.5 in 2nd year, then increase of 1 per year
33		Discounting factor	0.965	0.888	0.807	=1/(1+$C$2)^C32
34		**Present value of FCF**	**3,055**			=SUMPRODUCT(C31:E31,C33:E33)
35						

First, the free cash flow needs to be adjusted so that we use only three-quarters of the year's value, since the first quarter is not "historical" and is no longer part of the forecast flows for the DCF (row 25).

15.4.4.1 Year-end discounting

The year-end discounting shows the year count (row 26). The flows recognized at December 31, 2021, is discounted three-quarters of the year, so that's 0.75 years. The flows recognized at the end of the next year are discounted 0.75 + 1 (or 1.75) years, and the next one after that at 0.75 + 1 + 1 (2.75) years.

15.4.4.2 *Midyear (or midperiod) discounting*

The midyear discounting (and we should say midperiod discounting) is a little tricky. The free cash flow does not need any adjustments, but what is the year count?

Since we assume that the flows are recognized at the *middle* of the three-quarters of the first year that is left in the forecast, the year count is 0.75/2, or 0.375 (row 32).

For the second year, the flow is recognized at a point that is at the middle of the year in 2022 from March 31, 2021, so the year count is 0.5 + the 0.75 of the first year, or 1.25.

For the third year, the year count is 0.5 for the third year + 1.0 for the full second year and 0.75 of the first year, or 2.25. Subsequent years after this go up by 1.0 for each year.

15.4.5 Another look at discounting factors

Figure 15-4 shows the table of numbers in the model for the calculation of discount factors when the valuation date is March 31, 2021. The period-end count of years is straightforward. It's a 0.75 factor for the current year, going up by 1 in each successive year. The midperiod count uses half of the 0.75 for the first year, and then uses the full first year factor of 0.75 + 0.50 as the midpoint of that year. Thus, we get a 1.25 discount year number. From this point on, the number goes up by 1 for each year.

FIGURE 15-4

Discounting factors calculation in the model for a March 31, 2021 valuation

	A	B	C D E F	G	H	I	J	K
62	N	*Discounting factor to 31-Mar-21*		**Dec-21**	**Dec-22**	**Dec-23**	**Dec-24**	**Dec-25**
63	N	*Period-end year count*		*0.750*	*1.75*	*2.75*	*3.75*	*4.75*
64	N	*Mid-period year-count*		*0.375*	*1.25*	*2.25*	*3.25*	*4.25*

15.5 TERMINAL VALUE

Terminal value (TV) is the estimate of the value of the entity from the end of the projected period to perpetuity. It is based on the assumed growth into perpetuity of the FCF from the last projected year onward. This approach is called the *Gordon growth* method. The TV can also be derived by using an *exit multiple* of EBITDA. In our model, we have the option to do both. A net income multiple would not be appropriate, because that is a flow after interest expense and thus net income represents the flows available to the equity holders only. The basis of DCF is to use flows available to both equity and debt holders.

Since a DCF relies on a limited time horizon of 5 to 10 years, the terminal value represents the value of all the free cash flows extending from beyond the last projected year into perpetuity, since we assume that any healthy business will continue forever. Another way to think of the TV is to consider it as the price at which a buyer would buy the entity at the last forecast year.

Assumptions regarding the TV should be carefully developed. The terminal value is a large number and can be in the region of 60% to 85% of the enterprise value for companies that are mature. For startups with no immediate cash flow growth, or for companies where there is a much longer gestation period for projects, such as mining companies where it may take several years before a new mining operation becomes profitable, the forecast period should be long enough to include years of full operation so that the bulk of a company's valuation is not coming from the terminal value assumptions.

The TV is recognized at the last projected year and not at the last projected year + 1 year, as implied by the use of (1 + terminal growth) in the formula. It is discounted back to the valuation date at the rate of the WACC.

15.5.1 A "normalized" last forecast year for the calculation of terminal value

The simple way to develop the terminal value is to use the last projected year (shown in Figure 15-5 as Dec-25 in column K) and apply the growth formula or the exit multiple on the FCF or EBITDA, respectively. However, for the Gordon growth case, we should take into account two subtleties on the numbers going into perpetuity. For this reason, we should use a "normalized" last projected year that includes adjustments as described below. These are shown boxed in Figure 15-5. (These adjustments are not required for the EBITDA exit multiple, since EBITDA is not affected by depreciation and the change in net working capital.)

FIGURE 15-5

The normalized last forecast year

	A	B	C D E F	G	H	I	J	K	L	M
22		**WACC**		7.1%						**Normalized**
23				Proj	Proj	Proj	Proj	Proj		Proj
24		**DISCOUNTED CASH FLOW**		Dec-21	Dec-22	Dec-23	Dec-24	Dec-25		Dec-25
25		Revenue		3,150.0	3,307.5	3,472.9	3,646.5	3,828.8		3,828.8
26		**EBITDA**		787.5	826.9	868.2	911.6	957.2		957.2
27		Depreciation		(110.0)	(120.0)	(130.0)	(140.0)	(150.0)		(100.0)
28		Amortization		(30.0)	(30.0)	(30.0)	(30.0)	(30.0)		0.0
29		**EBIT**		647.5	676.9	708.2	741.6	777.2		857.2
30		Tax		(259.0)	(270.8)	(283.3)	(296.7)	(310.9)		(342.9)
31		Tax rate		40.0%	40.0%	40.0%	40.0%	40.0%		40.0%
32		**EBIAT**		388.5	406.1	424.9	445.0	466.3		514.3
33										
34		Depreciaton		110.0	120.0	130.0	140.0	150.0		100.0
35		Amortization		30.0	30.0	30.0	30.0	30.0		0.0
36		(Incr) / decr in net working capital		(106.9)	8.7	(63.4)	(45.6)	(62.9)		(36.2)
37		Capital expenditures		(100.0)	(100.0)	(100.0)	(100.0)	(100.0)		(100.0)
38		Free cash flow		321.6	464.8	421.5	469.4	483.5		478.1
39		Less: Pre-valuation date flows		(80.4)						
40		**Free cash flow for discounting**		241.2	464.8	421.5	469.4	483.5		**478.1**
41		**Terminal Value**								**12,011.1**

15.5.1.1 Capital expenditures and depreciation

Capital expenditures and depreciation are different during the forecast years. However, if we consider that each year's capital expenditure is fully depreciated over time, we can make the assumption for the perpetuity period that each capital expenditure ultimately equals its total depreciation. With this reasoning, for the normalized period, we can make the depreciation equal to the capital expenditure.

15.5.1.2 Amortization becomes 0

As the amortization of intangibles is limited to the starting amount of intangibles, we cannot assume that it will continue into perpetuity. Accordingly, we use 0 as the amortization for the normalized year going into perpetuity even if there are still unamortized intangibles on the balance sheet at the end of the forecast period.

15.5.1.3 Change in net working capital is based on the terminal growth rate

The terminal growth is typically a lower and more conservative value compared to the revenue growth in the forecast years. To the extent it is, the normalized last forecast year should have a change in the NWC that reflects the revenue growth for the terminal period. Rather than calculating the full working capital assets and liabilities, we can just take the NWC in the second-to-last forecast year (the year 2024) and apply the terminal growth rate to that. As before, we have to keep in mind that the increase should be shown as a negative number to designate that it is a use of cash. For the terminal value, the change in the NWC should never be a source of cash (i.e., a positive number) as that would imply that the company is reducing its NWC below the limits on the balance sheet.

This adjustment is important only for the Gordon growth TV setting. It does not apply in the exit multiple as that approach uses the EBIT number, which does not include NWC changes.

15.5.2 Year-end or midyear discounting for terminal value

A terminal growth TV is discounted using the same approach as the free cash flows.

However, a terminal growth using an exit multiple of EBITDA is always discounted using the year-end convention, regardless of the convention that is used to discount the FCF. This is because the multiples are based on year-end values for EBITDA, so the exit multiple value is a year-end value.

15.6 ENTERPRISE VALUE

The enterprise value is the total of the present value of the free cash flows and the terminal value. It represents the value of the entity to both the equity and debt holders and can be said to be the value of the whole company.

15.7 EQUITY VALUE

The equity value is the enterprise value less the market value of debt plus the cash on the balance sheet. This represents the value to the equity holders. If you divide the equity value by the number of shares outstanding, you get to the equity value per share. You can compare this value to the market price of the entity's shares, if they are publicly traded, and from this you can get a sense of whether the market overvalues or undervalues the entity.

15.8 CHALLENGES

DCF analysis relies on projected numbers (there is no such thing as a DCF using historical numbers), so getting the projections right is important. There are several areas that require attention:

- Develop the projections with due care in defining the drivers in the income statement from the revenue down to EBIT, the assumptions for the working capital, and the estimates for capital expenditures. These are the components for the free cash flows. The final result is only as good as the quality of the inputs.

- Have a time horizon in the projections of at least five years, or the length of a business cycle. A longer time horizon also allows you to model the company toward a "steady state" as it reaches toward the last projected year, with no major fluctuations in its operating flows or investment requirements in working capital and capital expenditures. The danger of a company that is not in a steady state is that the last projected year may present an upswing or downswing year. The TV will then be overestimated in the former and underestimated in the latter.
- Get the correct estimate for the WACC. The illustration in the model uses a simple WACC calculation, but you might consider other adjustments for company-specific and country risk. The size of the company also affects risk, with smaller companies (in terms of market capitalization) adding to the cost of equity. Very large companies have a size premium that reduces the cost of equity. The subtleties of deriving the correct WACC is beyond the scope of this chapter.

15.9 CONTINUING THE MODEL

The illustrations that follow are from the model from Chapter 11 and show a valuation model with the components typically seen in a DCF analysis. This layout is just one example that can be used for a DCF analysis module.

This DCF sheet uses the output results from the model that we have been developing in earlier chapters; the inputs for the free cash flow look to the output rows in that model. Of course, this DCF model could be built as a stand-alone module on its own, with the free cash flow entries reading inputs from other locations. As before, each figure is a screen shot, followed by a table that lists the formula entries in the cells shown in the illustration. The cells marked with the fine dotted lines are the input cells.

In the layout of this DCF sheet, you may notice that there are four columns (Columns C, D, E, and F) that appear to have no particular use. These columns are spacer columns so that the first year's column on this DCF sheet is aligned with the first projected year's column on the other sheets. This makes it easier to check for errors in the referencing across the sheets.

Figure 15-6 shows the inputs for the DCF. For ease of reference, I have put this box above the actual DCF layout.

The "N" seen in column A uses the row filtering techniques that I described earlier in Chapter 11, Section 11.12.2.1. I have also put in a Print button that automatically filters out the "N" lines and shows a preview of the page to print.

FIGURE 15-6

The DCF valuation input block

	A	B	G	H	I	K
1		**First Corporation**				
2		**Discounted cash flow**	**Scenario: Base**			**Print**
3		**in US$000**				
4						
5						
6	N	**Valuation date**	31-Mar-21		**Discounting**	Mid-period
7	N					
8	N	Risk-free rate	2.0%		Debt	979.7
9	N	+ (Equity risk premium	6.0%		Total capital	2,955.9
10	N	x Beta relevered)	1.10			
11	N	**= Cost of equity**	**8.6%**			
12	N				**Terminal value by**	Growth
13	N	Cost of debt	6.0%		Terminal growth	3.0%
14	N	Tax rate	40.0%		EBITDA exit multiple	4.0 x
15	N	**After-tax cost of debt**	**3.6%**			
16	N					
17	N	**Debt to capital ratio**	33.0%			
18	N					
19	N	**WACC**	**7.0%**			
20	N					

15.9.1 The valuation date

The starting point is the valuation date (cell G6), which defines the point in time to which the flows will be discounted.

15.9.2 Calculation of the WACC

The WACC calculation is as shown in Table 15-1:

TABLE 15-1

Cost of equity	Risk-free rate + (Beta × equity risk premium)
After-tax cost of debt	Cost of debt × (1 – tax rate %)
WACC	Cost of equity × % of equity in capital structure + After-tax cost of debt × % of market-value debt in capital structure

The 10-year U.S. Treasury bond rate is often used as representative of the risk-free rate. Be sure to check with your colleagues about which rate is the appropriate one to use as the risk-free rate and the equity risk premium. There are standards for these rates, and they can differ significantly from firm to firm.

Beta is a measure of the sensitivity of the movement of the company's stock price to the stock market in general. A beta of 1.0 means that the company's price moves in exact correlation with the market. A beta of 1.2 means that it is theoretically 20% more volatile than the market. Beta measures are derived by using regression analysis on the company's stock price sampled by daily, weekly, or monthly observations over a given time period.

15.10 A WALK THROUGH THE FORMULAS: THE MAIN DCF SECTION

Figure 15-7 and Table 15-2 show the formulas in the main DCF section. Shown are the formulas for the *last* forecast year (these formulas can be copied to the left to the start of the projections) and the normalized values for the same. The intent is to show the differences between these two columns that carry the same date and revenue and EBITDA numbers but begin to have differences below the EBITDA line.

FIGURE 15-7

Formulas for the main DCF section

	A	B	C D E F	N	O
24		**DISCOUNTED CASH FLOW**		**Column K, last forecast year**	**Column M, Terminal period**
25		Revenue		=Report!K7	=Report!K7
26		**EBITDA**		=Report!K12	=Report!K12
27		Depreciation		=Report!K14	=M37
28		Amortization		=Report!K15	0.0
29		**EBIT**		=K26+SUM(K27:K28)	=M26+SUM(M27:M28)
30		Tax		=-K29*K31	=-M29*M31
31		Tax rate		=Assumptions!K41	=Assumptions!K41
32		**EBIAT**		=SUM(K29:K30)	=SUM(M29:M30)
33					
34		Depreciaton		=-K27	=-M27
35		Amortization		=-K28	=-M28
36		(Incr) / decr in net working capit		=Report!K91	=IF($K$12="Growth",-N7*I72,Report!K91)
37		Capital expenditures		=Report!K97	=Report!K97
38		Free cash flow		=K32+SUM(K34:K37)	=M32+SUM(M34:M37)
39		Less: Pre-valuation date flows		=-(1-DAYS360(G6,G24)/360)*G38	**<== Column G only**
40		**Free cash flow for discounting**		=K38	=M38
41		**Terminal Value**			=IF($K$12="Growth",M40*(1+N7)/(G22-N7),M26*N7)
42					
43	N	*Discount factor*		=1/((1+$G$22)^K65)	=IF(K12="Growth",1/((1+$G$22)^K65),1/((1+$G$22)^K63))

TABLE 15-2

Row	Column B	Column K Dec 2025	Column M Normalized Dec 2025
25	Revenue	=Report!K7	=Report!K7. No difference.
26	EBITDA	=Report!K12	=Report!K12. No difference.
27	Depreciation	=Report!K14	=M37. This formula reads the capital expenditures.
28	Amortization	=Report!K15	This is 0. This is forced to be 0 as over time we expect that all intangibles will have been amortized.
29	EBIT	=K26+SUM(K27:K28)	=M26+SUM(M27:M28). Summing of the same three rows.
30	Tax	=–K29*K31	=–M29*M31. Same formula function, but different results because of the differences in the line above.
31	Tax rate	=Assumptions!K41	=Assumptions!K41. No difference.
32	EBIAT	=SUM(K29:K30)	=SUM(M29:M30). Same formula function, but different results because of the differences in the line above.
34	Depreciation	=–K27	=–M27
35	Amortization	=–K28	=–M28
36	(Incr) / decr in net working capital	=Report!K91	=IF($K$12="Growth",–N7*J72, Report!K91). Under the growth setting, the formula calculates the increase based on the terminal growth assumptions. Otherwise (i.e., under the EBITDA exit multiple), it uses the same formula.
38	Free cash flow	=K32+SUM(K34:K37)	=M32+SUM(M34:M37)

(continued on next page)

TABLE 15-2, *continued*

Row	Column B	Column K Dec 2025	Column M Normalized Dec 2025
39	Less: Pre-valuation date flows	**Column G only:** =−(1−DAYS360(G6,G24)/360)*G38. This is a calculation to remove the YTD portion of the free cash flow. The DAYS360(G6,G24) shows the number of days between the valuation date and the end of the first forecast year on a 360-day basis. Divide this by 360 and we get the portion of the year after the valuation date. Put the (1−... in front, and we get the portion of the year before the valuation date.	
40	Free cash flow for discounting	=K38	=M38
41	Terminal value	[Blank]	=IF($K$12="Growth",M40*(1+N7)/(G22−N7),M26*N7). If the TV setting is for "Growth," then this is the standard Gordon growth formula; otherwise (meaning we are using an EBITDA exit multiple), it multiplies the last forecast year's EBITDA with the multiple entry.
42	[Blank]		
43	Discount factor	=1/((1+$G$22)^K65)	=IF(K12="Growth",1/((1+$G$22)^K65), 1/((1+$G$22)^K63)). With the "Growth" setting, the same discount factor calculation is made. Row 65 contains the midyear or year-end exponent factor based on user inputs. Under the EBITDA multiple setting, the formula always uses the year-end factor.

15.11 A WALK THROUGH THE FORMULAS: WINDING UP

This section covers the formulas for the enterprise and equity values (Figures 15-8 and 15-9 and Table 15-3). The model has a data table to provide a range on the enterprise value based on sensitivities of WACC and either the terminal growth rate in increments of 0.25%, or the EBITDA exit multiple in increments of 1.0x.

FIGURE 15-8

The enterprise and equity value

	A	B	C	D	E	F	G	H
45		**Present value of free cash flows**					1,765.6	15.9%
46		**Present value of Terminal Value**					9,370.2	84.1%
47		**Enterprise value**					11,135.8	100.0%
48		Less: Debt					(979.7)	
49		Plus: Cash					0.0	
50		**Equity value**					10,156.1	
51								

FIGURE 15-9

Formulas for enterprise, equity value, and data table

	A	B	C	D	E	F	G	H	N
45		**Present value of free cash flows**					1,765.6	15.9%	=SUMPRODUCT(G40:K40,G43:K43)
46		**Present value of Terminal Value**					9,370.2	84.1%	=M41*M43
47		**Enterprise value**					11,135.8	100.0%	=SUM(G45:G46)
48		Less: Debt					(979.7)		=-SUM(Report!G50,Report!G55:G57)
49		Plus: Cash					0.0		=Report!G38
50		**Equity value**					10,156.1		=SUM(G47:G49)
51									

TABLE 15-3

Row	Column B	Column G
45	PV of free cash flows	=SUMPRODUCT(G40:K40,G43:K43). The SUMPRODUCT multiplies FCF elements in one range with the counterpart discounting factor elements in the other. This is a quick way of deriving each year's discounted values and summing them up.
46	PV of terminal value	=M41*M43. This multiplies the terminal value at the normalized last forecast year with the discount factor for that year. Note that the terminal value derived in the Gordon growth formula appears to be a value dated in the year after the last forecast year because there is a x (1 + growth rate%) in the formula, but it is discounted from the last forecast year point in time.

(continued on next page)

TABLE 15-3, *continued*

Row	Column B	Column G
47	Enterprise value	=SUM(G45:G46). The valuation number for the whole company.
48	Less: Debt	=–SUM(Report!G50,Report!G55:G57). This is a reference from the Report worksheet for all the debt, including any revolver. In a more detailed model, you should take care to include any short-term debt as well as the current portion of long-term debt.
49	Plus: Cash	=Report!G38. This is a direct reference to the excess cash number in the "Report" worksheet. In a more detailed model, this should include any "minimum cash" (a separate line item from excess cash) and also liquid short-term investments.
50	Equity value	=SUM(G47:G49). This is a sum because the debt is represented as a negative value in the SUM. In this case, because the cash is more than the debt, the equity value is more than the enterprise value. This currently shows the value to all equity holders, including minority interest and preferred stock holders. If you want to get to the common equity value, subtract the minority interest and preferred equity (we don't have this in our model) from the equity value.

15.12 THE DATA TABLE

Data tables allow multiple results from a single set of calculations. There are two types of data tables. In the one-variable type, we can vary one variable to see the results across many outputs. In the two-variable type, we can vary two variables and see the results of one output in permutations of the two inputs. The second one is the one that we are using here.

Data tables must be created (*Data>Forecast* section>*What-If Analysis>Data Table*) in the same worksheet as the calculations. This is one limitation. Of course, once you have a data table, you can reference it on another spreadsheet with the same layout. In this way, you are free to present the data tables on other worksheets for greater presentation impact.

15.12.1 What happens in a data table run?

Figure 15-10 shows the setup for the data table, the results of which will appear in a three-by-three grid for the nine scenario results based on a data table run during which Excel successively changes one of the variables in the calculations while holding the other unchanged. (This change is done internally in Excel; we would not see any changes in the inputs during the run.) It then displays the results in the grid, each in the intersection of the variable combination for that result. These actions are shown by the dashed lines in Figure 15-10.

FIGURE 15-10

The data table run



The grid is typically set up with the center value on each side as the starting or base case. As shown, the DCF has a WACC of 10% and a Gordon growth rate of 3.0%. On each side of these numbers in the grid, the values have plus-and-minus variances.

The block of calculations is our DCF module. When we set up the two-variable grid, it's actually a four-by-four grid because the top row and the leftmost column specify the range of variables to test. Let's call the top row the *x-row* (with the Gordon growth percentages) and the left column going down the *y-column* (with the WACC variances). The top left corner of the grid (shaded) is a direct Excel link from the output of the calculations. The x-row of the grid is for the sensitivities of one variable, in this case the terminal growth rate. The side of the grid, the y-column, is for WACC. These three components—the top left corner, the x-row, and the y-column—drive the data table run.

When done correctly, the setup for the three-by-three grid should produce at the center of the grid the value from the main calculation (shown bolded at the intersection of 3.0% growth and 10.0% WACC).

15.12.2 One big no-no when setting up a data table

One thing you must absolutely avoid: the values in the x-row and the y-column cannot directly refer to the variable cells in the calculations that Excel will use for the data run. The variable cells are the ones you selected when you were defining the two-variable data table (Figure 15-11). The row-input cell is the one that drives the x-row in the grid, so that would be the terminal growth assumption cell. The column-input cell drives the y-column, so that's the WACC cell.

If you do link these numbers, you are creating that dreaded circular reference: as Excel changes the calculation inputs based on the border ranges in the data table, those ranges are also changing because they are reading the permutations that Excel is making.

FIGURE 15-11

Defining the row and column input cells



It looks like whenever we create a data table, we have to manually set the range for the x-row and the y-columns. But wait, there is a way to get around this problem.

15.12.3 How to set the data table grid to follow the settings

Recall that Excel goes into the DCF calculations to vary the inputs for the two variables in order to create a data table. To avoid the circular reference of the x-row and y-column reading those inputs, create another input area before the data run calculation (Figure 15-12).

The x-row and y-column read this new input area. The new input area feeds the calculation block. Excel's data run sequences change the numbers in the calculation block but not anything in the input area. The numbers in the calculation block follow the sequential changes made by Excel to produce the results for the data table. In this way, you are separating the inputs you entered from the changes Excel is making creating the data tables.

FIGURE 15-12

Creating an input area separate from the data table actions



15.12.4 Two settings and two formats for the x-row

This section can be skipped if the model you are building only has one setting for the terminal value.

We have two settings for the terminal value: the Gordon growth and the EBITDA exit multiple. The x-row therefore has to change to either one of these. For the data table, whether the row is 0.03 (for 3% growth) or 4.0 (for 4.0 times EBITDA) is a matter of using an IF statement to pick the right input. However, for display purposes, it would be nice to have the first appear as 3.0% and the second as 4.0x. A macro can do it, but that makes it unnecessarily complex. We can use the TEXT function to format

FIGURE 15-11

Defining the row and column input cells



It looks like whenever we create a data table, we have to manually set the range for the x-row and the y-columns. But wait, there is a way to get around this problem.

15.12.3 How to set the data table grid to follow the settings

Recall that Excel goes into the DCF calculations to vary the inputs for the two variables in order to create a data table. To avoid the circular reference of the x-row and y-column reading those inputs, create another input area before the data run calculation (Figure 15-12).

The x-row and y-column read this new input area. The new input area feeds the calculation block. Excel's data run sequences change the numbers in the calculation block but not anything in the input area. The numbers in the calculation block follow the sequential changes made by Excel to produce the results for the data table. In this way, you are separating the inputs you entered from the changes Excel is making creating the data tables.

FIGURE 15-14

Formulas in columns I, J, and K

	Column I
Row 53	=IF($K$12="Growth",TEXT(I54,"0.0%"),TEXT(I54,"0.0 x"))
Row 54	=IF($K$12="Growth",J54-0.25%,J54-1)

	Column J
Row 53	=IF($K$12="Growth",TEXT(J54,"0.0%"),TEXT(J54,"0.0 x"))
Row 54	=IF($K$12="Growth",K13,K14)

	Column K
Row 53	=IF($K$12="Growth",TEXT(K54,"0.0%"),TEXT(K54,"0.0 x"))
Row 54	=IF($K$12="Growth",J54+0.25%,J54+1)

15.12.5 The look in the printout

Here is what the data table looks like in the final printout with the two TV settings (Figure 15-15).

FIGURE 15-15

How the data table looks in the printout with either TV setting

Enterprise value		Terminal growth %				
		2.5%	2.8%	3.0%	3.3%	3.5%
	6.5%	11,418	12,030	12,730	13,538	14,484
	6.7%	10,745	11,276	11,879	12,569	13,366
WACC	7.0%	10,147	10,612	11,136	11,730	12,410
	7.2%	9,613	10,023	10,481	10,997	11,583
	7.5%	9,133	9,496	9,900	10,352	10,860

Enterprise value		EBITDA exit multiple				
		2.0 x	3.0 x	4.0 x	5.0 x	6.0 x
	6.5%	3,208	3,919	4,631	5,342	6,053
	6.7%	3,182	3,886	4,589	5,293	5,996
WACC	7.0%	3,157	3,853	4,548	5,244	5,940
	7.2%	3,132	3,820	4,508	5,196	5,884
	7.5%	3,107	3,787	4,468	5,148	5,829

Note that row 54 is hidden by the filter. Overall, this is a polished layout regardless of the terminal value setting that is being used.